

计算机组成原理 实验报告

姓名：张艺耀

学号：PB20111630

实验日期：2022- -

实验题目

单周期CPU设计

实验目的

理解CPU的结构和工作原理

掌握单周期CPU的设计和调试方法

熟练掌握数据通路和控制器的设计和描述方法

实验平台

FPGAOL

Vivado

Mac + VSCode-remote + SSH + VLab

实验过程

Step 1 : 实现单周期RISC-V CPU 可执行以下十条命令(add addi sub auipc lw sw beq blt jal jalr)

本CPU采用与Ripes基本相同的数据通路 添加了多个mux和io_bus debug_bus。

top.v

```
1 module top(  
2     input clk,  
3     input rst,  
4  
5     //选择CPU工作方式  
6     input run,  
7     input step,  
8  
9     //输入switch的端口  
10    input valid,  
11    input [4:0] in,  
12  
13    //输出led和seg的端口  
14    output [1:0] check, //led6-5:查看类型  
15    output [4:0] out0,   //led4-0  
16    output [2:0] an,     //8个数码管  
17    output [3:0] seg,  
18    output ready        //led7  
19  
20 );  
21 wire clk_cpu;  
22  
23 wire [7:0] io_addr;  
24 wire [31:0] io_dout;  
25 wire io_we;  
26 wire [31:0] io_din;  
27  
28     wire [7:0] m_rf_addr;  
29 wire [31:0] rf_data;  
30 wire [31:0] m_data;  
31     wire [31:0] pc_out;  
32  
33 pdu pdu(clk, rst, run, step, clk_cpu, valid, in, check, out0, an, seg, ready, io_addr,  
34         io_dout, io_we, io_din, m_rf_addr, rf_data, m_data, pc_out);  
35  
36 cpu cpu(clk_cpu, rst, io_addr, io_dout, io_we, io_din, m_rf_addr, rf_data, m_data,  
37         pc_out);  
38  
39 endmodule
```

pdu.v

采用提供的源码。

```
1 module pdu(  
2     input clk,  
3     input rst,  
4  
5     //选择CPU工作方式  
6     input run,
```

```

7   input step,
8   output clk_cpu,
9
10  //输入switch的端口
11  input valid,
12  input [4:0] in,
13
14  //输出led和seg的端口
15  output [1:0] check, //led6-5:查看类型
16  output [4:0] out0, //led4-0
17  output [2:0] an, //8个数码管
18  output [3:0] seg,
19  output ready, //led7
20
21  //IO_BUS
22  input [7:0] io_addr,
23  input [31:0] io_dout,
24  input io_we,
25  output [31:0] io_din,
26
27  //Debug_BUS
28  output [7:0] m_rf_addr,
29  input [31:0] rf_data,
30  input [31:0] m_data,
31  input [31:0] pc
32 );
33
34 reg [4:0] in_r; //同步外部输入用
35 reg run_r, step_r, step_2r, valid_r, valid_2r;
36 wire step_p, valid_pn; //取边沿信号
37
38 reg clk_cpu_r; //寄存器输出CPU时钟
39 reg [4:0] out0_r; //输出外设端口
40 reg [31:0] out1_r;
41 reg ready_r;
42 reg [19:0] cnt; //刷新计数器，刷新频率约为95Hz
43 reg [1:0] check_r; //查看信息类型，00-运行结果，01-寄存器堆，10-存储器，11-PC
44
45 reg [7:0] io_din_a; //_a表示为满足组合always描述要求定义的，下同
46 reg ready_a;
47 reg [4:0] out0_a;
48 reg [31:0] out1_a;
49 reg [3:0] seg_a;
50
51 assign clk_cpu = clk_cpu_r;
52 assign io_din = io_din_a;
53 assign check = check_r;
54 assign out0 = out0_a;
55 assign ready = ready_a;
56 assign seg = seg_a;
57 assign an = cnt[19:17];
58 assign step_p = step_r & ~step_2r; //取上升沿

```

```

59 assign valid_pn = valid_r ^ valid_2r; //取上升沿或下降沿
60 assign m_rf_addr = {{3{1'b0}}, in_r};
61
62 //同步输入信号
63 always @(posedge clk) begin
64     run_r <= run;
65     step_r <= step;
66     step_2r <= step_r;
67     valid_r <= valid;
68     valid_2r <= valid_r;
69     in_r <= in;
70 end
71
72 //CPU工作方式
73 always @(posedge clk, posedge rst) begin
74     if(rst)
75         clk_cpu_r <= 0;
76     else if (run_r)
77         clk_cpu_r <= ~clk_cpu_r;
78     else
79         clk_cpu_r <= step_p;
80 end
81
82 //读外设端口
83 always @* begin
84     case (io_addr)
85         8'h0c: io_din_a = {{27{1'b0}}, in_r};
86         8'h10: io_din_a = {{31{1'b0}}, valid_r};
87         default: io_din_a = 32'h0000_0000;
88     endcase
89 end
90
91 //写外设端口
92 always @(posedge clk, posedge rst) begin
93     if (rst) begin
94         out0_r <= 5'h1f;
95         out1_r <= 32'h1234_5678;
96         ready_r <= 1'b1;
97     end
98     else if (io_we)
99         case (io_addr)
100             8'h00: out0_r <= io_dout[4:0];
101             8'h04: ready_r <= io_dout[0];
102             8'h08: out1_r <= io_dout;
103             default: ;
104         endcase
105     end
106
107 //LED和数码管查看类型
108 always @(posedge clk, posedge rst) begin
109     if(rst)
110         check_r <= 2'b00;

```

```

111     else if(run_r)
112         check_r <= 2'b00;
113     else if (step_p)
114         check_r <= 2'b00;
115     else if (valid_pn)
116         check_r <= check - 2'b01;
117 end
118
119 //LED和数码管显示内容
120 always @* begin
121     ready_a = 1'b0;
122     case (check_r)
123         2'b00: begin
124             out0_a = out0_r;
125             out1_a = out1_r;
126             ready_a = ready_r;
127         end
128         2'b01: begin
129             out0_a = in_r;
130             out1_a = rf_data;
131         end
132         2'b10: begin
133             out0_a = in_r;
134             out1_a = m_data;
135         end
136         2'b11: begin
137             out0_a = 5'b00000;
138             out1_a = pc;
139         end
140     endcase
141 end
142
143 //扫描数码管
144 always @(posedge clk, posedge rst) begin
145     if (rst) cnt <= 20'h0_0000;
146     else cnt <= cnt + 20'h0_0001;
147 end
148
149 always @* begin
150     case (an)
151         3'd0: seg_a = out1_a[3:0];
152         3'd1: seg_a = out1_a[7:4];
153         3'd2: seg_a = out1_a[11:8];
154         3'd3: seg_a = out1_a[15:12];
155         3'd4: seg_a = out1_a[19:16];
156         3'd5: seg_a = out1_a[23:20];
157         3'd6: seg_a = out1_a[27:24];
158         3'd7: seg_a = out1_a[31:28];
159         default: ;
160     endcase
161 end
162

```

cpu.v

CPU主模块 为整个数据通路整体的框架。

```

1  module cpu(
2      input clk,
3      input rst,
4
5      //io_bus
6      output [31:0] io_addr,    //led和seg的地址
7      output [31:0] io_dout,    //输出led和seg的数据
8      output io_we,            //输出led和seg数据时的使能信号
9      input [31:0] io_din,      //输出led和seg数据时的使能信号
10
11     //debug_bus
12     input [7:0] m_rf_addr,    //存储器 (mem) 或寄存器堆 (rf) 的调试读口地址
13     output [31:0] rf_data,    //从rf读取的数据
14     output [31:0] m_data,     //从mem读取的数据
15     output [31:0] pc_out      //pc的内容
16 );
17
18 wire branch;
19 wire [31:0] pc;
20 wire [31:0] pc_in;
21 wire [31:0] pc_plus4;
22 wire [31:0] inst;
23 wire [31:0] imm_out;
24 wire rf_wr_en;
25 wire alu_a_sel;
26 wire alu_b_sel;
27 wire [1:0] alu_ctrl;
28 wire dm_rd_ctrl;
29 wire dm_wr_ctrl;
30 wire dm_wr_ctrl_aft;
31 wire [1:0] rf_wr_sel;
32 wire [2:0] comp_ctrl;
33 wire do_branch;
34 wire do_jump;
35
36 reg [31:0] rf_wd;
37 wire [31:0] rf_rd0, rf_rd1;
38 wire [31:0] alu_a, alu_b, alu_out;
39 wire [31:0] dm_dout;
40 wire [31:0] dm_dout_aft;
41
42 assign alu_a = alu_a_sel ? rf_rd0 : pc;
43 assign alu_b = alu_b_sel ? imm_out : rf_rd1;
44 assign pc_in = (pc >= 32'h3000 && pc <= 32'h33ff) ? pc : 32'h0;
45 assign pc_out = pc_in;

```

```

46
47 //io_bus
48 assign dm_wr_ctrl_aft = (~(io_addr[10])) & dm_wr_ctrl; //AND gate
49 assign dm_dout_aft = io_addr[10] ? io_din : dm_dout; //Mux
50 assign io_dout = rf_rd1;
51 assign io_we = io_addr[10] & dm_wr_ctrl; //Mux
52 assign io_addr = alu_out;
53
54 program_counter program_counter(clk, rst, branch, alu_out, pc, pc_plus4);
55 alu alu(alu_a, alu_b, alu_ctrl, alu_out);
56 register_file register_file(.clk (clk), .rst (rst), .ra0 (inst[19:15]), .ra1
    (inst[24:20]), .ra2 (m_rf_addr[4:0]), .wa (inst[11:7]), .wd (rf_wd), .we (rf_wr_en), .rd0
    (rf_rd0), .rd1 (rf_rd1), .rd2 (rf_data));
57 imm imm(.inst (inst), .imm_out (imm_out));
58 br br(.a (rf_rd0), .b (rf_rd1), .comp_ctrl (comp_ctrl), .do_branch (do_branch), .do_jump
    (do_jump), .branch (branch));
59 controller controller(inst, rf_wr_en, alu_a_sel, alu_b_sel, alu_ctrl, dm_rd_ctrl,
    dm_wr_ctrl, rf_wr_sel, comp_ctrl, do_branch, do_jump);
60
61 ins_mem ins_mem(.a (pc_in[9:2]), .spo (inst));
62 data_mem data_mem(.a (alu_out[9:2]), .d (rf_rd1), .dpra (m_rf_addr), .clk (clk), .we
    (dm_wr_ctrl_aft), .spo (dm_dout), .dpo (m_data));
63
64 //Mux
65 always@(*) begin
66     case (rf_wr_sel)
67         2'b00: rf_wd = 32'h0;
68         2'b01: rf_wd = pc_plus4;
69         2'b10: rf_wd = alu_out;
70         2'b11: rf_wd = dm_dout_aft;
71         default: rf_wd = 32'h0;
72     endcase
73 end
74
75 endmodule

```

program_counter.v

PC模块 rst为1时重置初始地址为 0x3000。

```

1 module program_counter(
2     input clk,
3     input rst,
4     input br,
5     input [31:0] alu_out,
6     output reg [31:0] pc,
7     output [31:0] pc_plus4
8 );
9
10 always@(posedge clk or posedge rst) begin
11     if(rst) pc <= 32'h3000;

```

```

12     else if(br) pc <= alu_out;
13     else pc <= pc_plus4;
14 end
15
16 assign pc_plus4 = pc + 32'h4;
17
18 endmodule

```

alu.v

ALU（算术逻辑运算单元） 由于本次实验的运算操作仅涉及到加减，故省略了逻辑操作和移位操作等。

```

1 module alu(
2     input [31:0] a,b,
3     input [1:0] alu_ctrl,
4     output reg [31:0] alu_out
5 );
6
7 always@(*) begin
8     case(alu_ctrl)
9         4'h0: alu_out = a + b;
10        4'h1: alu_out = a - b;
11        default: alu_out = 32'h0;
12    endcase
13 end
14
15 endmodule

```

register.v

三端口寄存器（端口3用于连接io总线访问外设）。

```

1 module register_file (
2     input clk, rst,
3     input [4:0] ra0,    //读端口0地址
4     output [31:0] rd0,  //读端口0数据
5
6     input [4:0] ra1,    //读端口1地址
7     output [31:0] rd1,  //读端口1数据
8
9     input [4:0] ra2,    //读端口2地址
10    output [31:0] rd2,  //读端口2数据
11
12    input [4:0] wa, //写端口地址
13    input we,    //写使能 高电平有效
14    input [31:0] wd //写端口数据
15 );
16
17 reg [31:0] regfile [0:31];
18 assign rd0 = ra0 ? regfile[ra0] : 32'h0;
19 assign rd1 = ra1 ? regfile[ra1] : 32'h0;

```



```

20 assign rd2 = ra2 ? regfile[ra2] : 32'h0;
21 integer i;
22
23 always @(posedge clk) begin
24     if(rst) begin
25         for(i = 0; i <= 31; i = i + 1) begin
26             regfile[i] <= 32'b0;
27         end
28     end
29     else if(we) begin
30         if(wa == 0) regfile[wa] <= 0;    //r0内容恒为0
31         else regfile[wa] <= wd;
32     end
33 end
34
35 endmodule

```

imm.v

立即数扩展模块。

```

1  module imm(
2      input [31:0] inst,
3      output reg [31:0] imm_out
4  );
5  wire [6:0] inst_type;
6
7  assign inst_type = inst[6:0];
8
9  always@(*) begin
10     case(inst_type)
11         7'h03: imm_out = {{21{inst[31]}}}, inst[30:20]]; //i_type:lw
12         7'h13: imm_out = {{21{inst[31]}}}, inst[30:20]]; //i_type:logic
13         7'h17: imm_out = {inst[31:12], 12'h0}; //aupic
14         7'h37: imm_out = {inst[31:12], 12'h0}; //lui
15         7'h6f: imm_out = {{13{inst[31]}}}, inst[19:12], inst[20], inst[30:21], 1'b0}; //jal
16         7'h63: imm_out = {{20{inst[31]}}}, inst[7], inst[30:25], inst[11:8], 1'b0}; //b_type
17         7'h23: imm_out = {{21{inst[31]}}}, inst[30:25], inst[11:7]]; //s_type
18         7'h67: imm_out = {{21{inst[31]}}}, inst[30:20]]; //jalr
19         default: imm_out = 32'h0;
20     endcase
21 end
22
23 endmodule

```

br.v

branch模块，根据控制模块输出comp_ctrl决定是否跳转。

```
1  module br(  
2      input  [31:0] a,  
3      input  [31:0] b,  
4      input  [2:0] comp_ctrl,  
5      input  do_branch,  
6      input  do_jump,  
7      output branch  
8  );  
9  reg taken;  
10  
11  always@(*) begin  
12      case(comp_ctrl)  
13          3'h0: taken = (a == b);  
14          3'h4: taken = (a < b);  
15          default: taken = 0;  
16      endcase  
17  end  
18  
19  assign branch = (taken && do_branch) || do_jump;  
20  
21  endmodule
```

controller.v

控制模块，解析指令类型并产生相应控制信号。

```
1  module controller(  
2      input  [31:0] inst,  
3      output rf_wr_en, alu_a_sel, alu_b_sel,  
4      output reg [1:0] alu_ctrl,  
5      output reg dm_rd_ctrl,  
6      output reg dm_wr_ctrl,  
7      output reg [1:0] rf_wr_sel,  
8      output [2:0] comp_ctrl,  
9      output do_branch, do_jump  
10 );  
11  
12 wire [6:0] opcode;  
13 wire [2:0] funct3;  
14 wire [6:0] funct7;  
15  
16 wire is_add;  
17 wire is_addi;  
18 wire is_sub;  
19 wire is_avipc;  
20 wire is_lw;  
21 wire is_sw;
```

```

22 wire is_beq;
23 wire is_blt;
24 wire is_jal;
25 wire is_jalr;
26
27 wire is_add_type;
28 wire is_u_type;
29 wire is_jump_type;
30 wire is_b_type;
31 wire is_r_type;
32 wire is_i_type;
33 wire is_s_type;
34
35 assign opcode = inst[6:0];
36 assign funct3 = inst[14:12];
37 assign funct7 = inst[31:25];
38
39 assign is_add = (opcode == 7'h33) && (funct3 == 3'h0) && (funct7 == 7'h0);
40 assign is_addi = (opcode == 7'h13) && (funct3 == 3'h0);
41 assign is_sub = (opcode == 7'h33) && (funct3 == 3'h0) && (funct7 == 7'h20);
42 assign is_avipc = (opcode == 7'h17);
43 assign is_lw = (opcode == 7'h03) && (funct3 == 3'h2);
44 assign is_sw = (opcode == 7'h23) && (funct3 == 3'h2);
45 assign is_beq = (opcode == 7'h63) && (funct3 == 3'h0);
46 assign is_blt = (opcode == 7'h63) && (funct3 == 3'h4);
47 assign is_jal = (opcode == 7'h6f);
48 assign is_jalr = (opcode == 7'h67) && (funct3 == 3'h0);
49
50 assign is_add_type = is_avipc | is_jal | is_jalr | is_b_type | is_s_type | is_lw | is_add
    | is_addi;
51 assign is_u_type = is_avipc;
52 assign is_jump_type = is_jal;
53 assign is_b_type = is_beq | is_blt;
54 assign is_r_type = is_add | is_sub;
55 assign is_i_type = is_jalr | is_lw | is_addi;
56 assign is_s_type = is_sw;
57
58 always@(*) begin
59     if(is_add_type) alu_ctrl = 2'h0;
60     else if(is_sub) alu_ctrl = 2'h1;
61 end
62
63 assign rf_wr_en = is_u_type | is_jump_type | is_r_type | is_i_type;
64 assign alu_a_sel = is_r_type | is_i_type | is_s_type;
65 assign alu_b_sel = ~ is_r_type;
66
67 always@(*) begin
68     if(is_lw) dm_rd_ctrl = 1'b1;
69     else dm_rd_ctrl = 1'b0;
70 end
71
72 always@(*) begin

```

```

73     if(is_sw) dm_wr_ctrl = 1'b1;
74     else dm_wr_ctrl = 1'b0;
75 end
76
77 always@(*) begin
78     if(opcode == 7'h3) rf_wr_sel = 2'h3;
79     else if(((~ is_jalr) & is_i_type) | is_u_type | is_r_type) rf_wr_sel = 2'h2;
80     else if(is_jal | is_jalr) rf_wr_sel = 2'h1;
81     else rf_wr_sel = 2'h0;
82 end
83
84 assign comp_ctrl = funct3;
85 assign do_branch = is_b_type;
86 assign do_jump = is_jal | is_jalr;
87
88 endmodule

```

ins_mem.v 和 data_mem.v

采用ip核例化，ins_mem选用25632ROM、data_mem选用25632的双端口RAM。

文章 <https://cloud.tencent.com/developer/article/1814213> 介绍了不同端口的分布式和块式RAM的区别。

Step 2 : 对10条指令进行仿真测试

在RARS中编写汇编测试文件（根据上一实验改写而成 新增了四条指令）

test.asm

```

1  .data
2  out: .word 0xff
3  in: .word 0xff
4
5  .text
6  la a0, out    #auipc
7  sw x0, 0(a0)  #sw 0 in address out
8  addi x5, x0, 0xf0 #addi
9  lw x5, 4(a0)  #lw
10
11 li x6, 0xf
12 li x7, 0xf0
13 add x5, x6, x7 #add
14 sub x5, x6, x7 #sub
15
16 li x7, 0xf
17 beq x6, x7, equal #beq
18 jal exit #jal
19

```

```

20 equal:
21 li x7, 0xf0
22 blt x6, x7, lessthan #blt
23 jal exit
24
25 lessthan:
26 la a0, exit
27 jalr a0 #jalr
28
29 exit:
30 li a7, 10
31 ecall

```

生成 `ins.coe` 和 `data.coe`

ins.coe

```

1 memory_initialization_radix = 16;
2 memory_initialization_vector =
3 ffffd517
4 00050513
5 00052023
6 0f000293
7 00452283
8 00f00313
9 0f000393
10 007302b3
11 407302b3
12 00f00393
13 00730463
14 01c000ef
15 0f000393
16 00734463
17 010000ef
18 00000517
19 00c50513
20 000500e7
21 00a00893
22 00000073

```

data.coe

```

1 memory_initialization_radix = 16;
2 memory_initialization_vector =
3 000000ff
4 000000ff

```

仿真

test_bench.v

首先查看 x5 寄存器值。

```
1  module test_bench();
2  reg clk;
3  reg rst;
4
5  wire [31:0] io_addr;
6  wire [31:0] io_dout;
7  wire io_we;
8  reg [31:0] io_din;
9
10 reg [7:0] m_rf_addr;
11 wire [31:0] rf_data;
12 wire [31:0] m_data;
13 wire [31:0] pc_out;
14
15 cpu cpu(clk, rst, io_addr, io_dout, io_we, io_din, m_rf_addr, rf_data, m_data, pc_out);
16
17 initial begin
18     rst = 1; m_rf_addr = 5;
19     clk = 0;#5
20     rst = 0;
21
22     forever #5 clk = ~clk;
23 end
24
25 initial begin
26     #800 $finish;
27 end
28
29 endmodule
```


fib.asm

```
1  .data
2  out: .word 0
3  in: .word 2
4
5  .text
6  li t1, 1
7  li t2, 2
8  la a0, out
9  lw t3, 4(a0)
10 beq t1, t3, first
11 beq t2, t3, second
12 addi t3, t3, -2
13 loop:
14 add t4, t1, t2
15 addi t1, t2, 0
16 addi t2, t4, 0
17 addi t3, t3, -1
18 bgtz t3, loop
19 sw t4, 0(a0)
20 jal exit
21
22 first:
23 sw t1, 0(a0)
24 jal exit
25
26 second:
27 sw t2, 0(a0)
28 jal exit
29
30 exit:
31 li a7, 10
32 ecall
```

fib_data.coe

```
1 memory_initialization_radix = 16;
2 memory_initialization_vector =
3 00000000
4 00000005
```


fib_ins.coe

```
1 memory_initialization_radix = 16;
2 memory_initialization_vector =
3 00100313
4 00200393
5 fffffd517
6 ff850513
```



```
7 00452e03
8 03c30463
9 03c38663
10 ffee0e13
11 00730eb3
12 00038313
13 000e8393
14 fffe0e13
15 ffc048e3
16 01d52023
17 014000ef
18 00652023
19 00c000ef
20 00752023
21 004000ef
22 00a00893
23 00000073
```

仿真波形如下：



`rf_data` 中读出的结果为8 正确。

实验结果

测试Lab3 fib.s

烧写到FPGAOL上 mem reg显示结果均与仿真结果一致。

led7 led6 led5 led4 led3 led2 led1 led0




G18 F18 E17 D17 G17 E18 D18 C17

FPGA

XC7A100t-CSG324-1

H16 G13 F13 E16 H14 G16 F16 D14




sw7 sw6 sw5 sw4 sw3 sw2 sw1 sw0

segplay(sharing with led)

hexplay




pc为0x3008时 寄存器x6的值为1



计算fib第五项结果存到data_mem[0]处 结果为8

FPGA interface



segplay(sharing with led) hexplay



测试fib_test.s (外设输入)

fib_test.asm

```
1 .text
2     addi x1, x0, 1    #x1=1
3     add t1, x0, x0    #store fib series @t1
4
5     #### input f0
6     sw x1, 0x404(x0)  #rdy=1
7 l1:
8     lw t0, 0x410(x0)  #wait vld=1
9     # addi a7, x0, 5   #for debug begin
10    # ecall
```

```

11 #     mv t0, a0           #for debug end
12
13     beq t0, x0, l1
14     lw s0, 0x40c(x0)      #s0=vin
15 #     addi a7, x0, 5       #for debug begin
16 #     ecall
17 #     mv s0, a0           #for debug end
18
19     sw s0, 0x408(x0)      #out1=f0
20     sw s0, 0(t1)          #store f0
21     addi t1, t1, 4
22
23     sw x0, 0x404(x0)      #rdy=0
24 l2:
25     lw t0, 0x410(x0)      #wait vld=0
26 #     addi a7, x0, 5       #for debug begin
27 #     ecall
28 #     mv t0, a0           #for debug end
29
30     beq t0, x1, l2
31
32 ##### input f1
33     sw x1, 0x404(x0)      #rdy=1
34 l3:
35     lw t0, 0x410(x0)      #wait vld=1
36 #     addi a7, x0, 5       #for debug begin
37 #     ecall
38 #     mv t0, a0           #for debug end
39
40     beq t0, x0, l3
41     lw s1, 0x40c(x0)      #s1=vin
42 #     addi a7, x0, 5       #for debug begin
43 #     ecall
44 #     mv s1, a0           #for debug end
45
46     sw s1, 0x408(x0)      #out1=f1
47     sw s1, 0(t1)          #store f1
48     addi t1, t1, 4
49
50     sw x0, 0x404(x0)      #rdy=0
51 l4:
52     lw t0, 0x410(x0)      #wait vld=0
53 #     addi a7, x0, 5       #for debug begin
54 #     ecall
55 #     mv t0, a0           #for debug end
56
57     beq t0, x1, l4
58
59 ##### comput fi = fi-2 + fi-1
60 next:
61     add t0, s0, s1        #fi
62     sw t0, 0x408(x0)      #out1=fi

```

```

63     sw t0, 0(t1)      #store fi
64     addi t1, t1, 4
65
66     add s0, x0, s1
67     add s1, x0, t0
68
69     sw x1, 0x404(x0)  #rdy=1
70 l5:
71     lw t0, 0x410(x0)  #wait vld=1
72 #     addi a7, x0, 5    #for debug begin
73 #     ecall
74 #     mv t0, a0         #for debug end
75
76     beq t0, x0, l5
77     sw x0, 0x404(x0)  #rdy=0
78 l6:
79     lw t0, 0x410(x0)  #wait vld=0
80 #     addi a7, x0, 5    #for debug begin
81 #     ecall
82 #     mv t0, a0         #for debug end
83
84     beq t0, x1, l6
85     jal x0, next

```

fib_test.coe

```

1  memory_initialization_radix = 16;
2  memory_initialization_vector =
3  00100093
4  00000333
5  40102223
6  41002283
7  fe028ee3
8  40c02403
9  40802423
10 00832023
11 00430313
12 40002223
13 41002283
14 fe128ee3
15 40102223
16 41002283
17 fe028ee3
18 40c02483
19 40902423
20 00932023
21 00430313
22 40002223
23 41002283
24 fe128ee3
25 009402b3


```

```

26 40502423
27 00532023
28 00430313
29 00900433
30 005004b3
31 40102223
32 41002283
33 fe028ee3
34 40002223
35 41002283
36 fe128ee3
37 fd1ff06f

```

FPGA interface



segplay(sharing with led) hexplay

00000007b

结果正确。

心得体会

本实验由于工程量较大，在cpu数据通路编写和debug方面花费了较多时间。中间由于部分端口打错，仿真与烧写结果不一致，对比二者后发现在top模块端口名打错。但是vivado在generatebitstream过程中并未弹出错误提示，因此这点应该多加小心。

建议是实验文档可以详细说一下pdu的功能。